

*Programmable
Sound
Generator*

TK 83/TK 85

MANUAL DE INSTRUÇÃO

Este gerador de sons programável (PSG- do inglês PROGRAMMABLE SOUND GENERATOR) é constituído pelo circuito integrado AY-3-8912. Trata-se de um sofisticado componente que possui três canais de saída. Programável em BASIC ou por Linguagem de Máquina, é capaz de criar uma enorme variedade de efeitos sonoros.

O PSG pode gerar notas musicais, com frequência e volume programáveis, além de possuir um gerador de ruído.

Após entender o mecanismo de funcionamento do PSG e experimentar os exemplos indicados, você estará em condições de executar vários efeitos espetaculares.

Inicialmente são apresentados alguns programas que demonstrarão o que pode ser obtido através do PSG e permitirão que você se familiarize com os modos de controle do PSG.

O primeiro programa apresentado abaixo gera sinais aleatórios (randômicos) para o PSG processá-los. A linha 1 em particular deve ser exatamente copiada como apresentado, usando-se os caracteres específicos indicados no teclado. Por exemplo, para 'PEEK' deve pressionar-se <SHIFT> junto com <NEW LINE> e a seguir a tecla 'O'. Os espaços em branco apresentados na linha 1 são gerados pelo próprio TK-85.

Após digitar o programa, pressione RUN. Alguns momentos passarão antes de o som começar.

```
1 REM YYPEEK TO YYPEEK ?TAN
2 GOTO 7
3 POKE A,D
4 POKE B,C
5 LET X=USR E
6 RETURN
7 LET A=16515
8 LET B=16519
9 LET E=16514
10 REM SAIDA RANDOMICA
20 LET C=INT (RND * 255)
30 LET D=INT (RND *14)
40 GOSUB 3
50 FOR T= 1 TO 10
60 NEXT T
70 GOTO 20
```

Você deverá notar que o PSG emitirá principalmente ruídos variados e somente poucos tons puros. Isto ocorre devido ao fato de que para o PSG produzir algum som puro, ou seja, emitir uma frequência sem harmônicos, são necessárias condições de controle específicas.

Agora experimente terminar o programa pressionando BREAK. A primeira coisa que você vai notar é que o PSG continuará a emitir o último ruído gerado antes de você ter acionado BREAK. E continuará a emitir este ruído até a UCP indicar outra ordem de comando para terminar a execução deste som.

O próximo programa irá habilitá-lo a comandar o PSG através do controle de duas variáveis (que são chamadas Destino (D) e Conteúdo (C)).

Listando o primeiro programa você notará que a linha 1 foi alterada! Não se preocupe em consertá-la, pois tudo está sob controle. De fato você irá notar que o segundo e quarto 'Y' serão alterados cada vez que você rodar o programa com o PSG. Este ponto voltará a ser discutido em detalhes mais adiante.

Mantenha as linhas 1 até 9 como estão e altere as linhas a partir da 10 mais o programa abaixo:

```
10 PRINT AT 0,0;"DIGITE O REGISTRO DESTINO"  
20 INPUT D  
30 SCROLL  
40 PRINT "D = ";D  
50 PRINT AT 0,0;"DIGITE O CONTEUDO"  
60 INPUT C  
70 GOSUB 3  
80 PRINT AT 21,10;"C = ";C  
90 GOTO 10
```

Os comandos INPUT's usados neste programa possibilitarão que você consiga sons variados e mostrarão como entrar com as variáveis para gerar sons.

Após pressionar RUN comece entrando com D=8, C=0; D=9, C=0; D=10, C=0. Estes comandos irão silenciar os três canais de som do PSG, tornando-o apto para algumas experiências que serão realizadas a seguir. Entre D=0, C=252, D=7, C=56 para habilitar os canais A, B e C. Agora entre D=8, C=15 (volume máximo no canal A). Você terá agora um tom simples estável saindo do alto falante. Entre agora com D=2, C=168; D=9 e C=15 (volume máximo no canal B), obtendo assim uma agradável combinação de duas notas. Agora faça D=4, C=89; D=10 e C=15 (volume máximo no canal C) para obter uma somatória dos três canais em volume máximo.

Agora o próximo passo será em alterar o volume de som do canal A, mantendo os canais B e C inalteráveis. Assim, entre com D=8, C=16 (para o controle de volume). Note que agora o canal A está desligado, enquanto o B e C se mantêm ligados e constantes. Entre com D=12 e C=64 (para regular a frequência de variação de volume na ordem de 2 segundos aproximadamente). Entrando com D=13 e C=0; você notará que o canal A surge e desaparece novamente.

Agora traga os três canais para o controle de volume: D=9, C=16; D=10, C=16 (silenciando os canais B e C). Entre agora com D=13 e C=0 outra vez. Todos os três canais aparecerão de uma vez e morrerão todos juntos. Tente D=13 e C=10 para uma sequência de subidas e descidas contínuas, ou D=13, C=12 para produzir apenas uma sequência crescente de volume (como uma forma de dente de serra). A seguir tente outras opções que o registro 13 pode proporcionar (veja a página 8).

Agora iremos desligar os geradores de tons e examinar o uso do ruído. Comece entrando com D=6 e C=31 (para produzir um ruído de baixa frequência). Depois entre D=7, C=7 para desativar os tons e emitir ruído nos três canais. Entre D=13 e C=0 para gerar uma simples emissão por vez, o que, no final, poderá ser sentido como se estivéssemos presenciando uma explosão. Agora, para uma rápida explosão, digite D=12 e C=16 e sinta o resultado, digitando D=13 e C=0.

Entre agora com D=12, C=0; D=11, C=128; D=13, C=10. Teremos então o semelhante ao som de um helicóptero. É interessante, neste estágio, experimentar as várias formas de saída para D=13, para encontrar o significado de cada entrada na tabela que será apresentada mais tarde.

Agora iremos verificar o efeito de misturar ruídos com tons em um mesmo canal e a seguir acrescentaremos um canal por vez e verificaremos os resultados. Faça D=9, C=0; D=10, C=0 desligando os canais B e C. Faça D=7, C=0 para habilitar tons e ruídos nos três canais. Depois faça D=8, C=15 gerando assim uma mistura de tons e ruídos no canal A.

Tendo mostrado como controlar o PSG ponto a ponto, a próxima seção irá descrever com muito mais detalhes o funcionamento do PSG, indicando como gerar sons.

O FUNCIONAMENTO DO PSG

O PSG contém três canais de sons, independentemente controláveis em frequência e em volume. Os canais são denominados A, B e C e suas operações são controladas pelas variáveis D e C. A variável D identifica um registrador interno ao PSG (existem 14 registradores), enquanto que a variável C especifica o conteúdo do registrador apontado por D. Cada registrador interno possui uma função específica que será descrita posteriormente.

Cada canal deve ser usado para gerar um simples tom cuja frequência deve ser acertada pelo usuário através da alteração do conteúdo de um par de registradores: 0 e 1 para o canal A, 2 e 3 para o canal B e 4 e 5 para o canal C. Por outro lado, cada canal pode ser usado para transmitir ruído, cuja frequência (ou nota) seja a mesma para os três canais, e seja regulada em um registrador separado (6).

O registrador 7 é usado para controlar o tipo de som de cada um dos três canais geradores, isto é, tons, ruídos ou ambos. Qualquer combinação pode ser selecionada inserindo o código apropriado no registrador 7.

O volume de cada canal pode ser controlado através dos registradores 8, 9 e 10, respectivamente para os canais A, B e C. Duas opções são disponíveis para este controle de volume: o fixo e o variável. No modo fixo existem 16 níveis disponíveis entre zero (silêncio) e 15 (máximo volume). No modo variável, o volume sempre varia entre o máximo e o mínimo nível. A velocidade com que o volume é variado e regulado pelos registros 11 e 12 será descrita na seção intitulada "Operações internas do PSG". Finalmente, o registro 13 controla o modo como o som varia. Licitos como sinos, bombas etc, ruídos repetitivos como helicópteros, sonares, podem todos serem determinados neste registrador.

O conteúdo carregado em qualquer registrador é mantido

até que um novo valor seja recarregado em cima do antigo. Assim sendo o valor de um registro só deve ser modificado onde uma alteração seja atualmente necessária. Efeitos como explosões, por exemplo, podem ser ativados por um simples carregamento no registro 13, para iniciar um novo efeito.

O CONTROLE ATRAVÉS DO TK-85

Antes de sabermos as funções de cada registro em detalhes e como eles atuam, é necessário tomarmos consciência da maneira com que o programa BASIC aciona o PSG.

Como já foi adiantado no item anterior, o PSG dispõe de duas entradas com 14 registros. Estes registros são carregados através de uma pequena rotina em Linguagem de Máquina na primeira linha do programa BASIC contido num comando REM, que deverá constar em todo programa em que você pretender utilizar o gerador de som.

Esta pequena rotina em Linguagem de Máquina deverá ser acompanhada por outras linhas de comando BASIC provocando o acesso rápido e fácil à rotina em Linguagem de Máquina. Para isso, nas linhas seguintes será apresentada uma sugestão para o início de qualquer programa utilizando o PSG.

```
1 REM YYPEEK TO YYPEEK ?TAN
2 GOTO 10
3 POKE 16515,D
5 POKE 16519,C
7 LET X=USR 16514
8 RETURN
10 REM SEU PROGRAMA COMEÇA AQUI
```

Note que esta pequena sub-rotina receberá os valores D e C e imediatamente os enviará ao PSG. Antes de se chamar esta sub-rotina, será sempre necessário carregar os registros D e C com seus novos valores, caso contrário o PSG não mostrará variação em seu comportamento.

O programa em Linguagem de Máquina do PSG deverá sempre ser chamado através de uma sub-rotina BASIC, porque ele necessitará frequentemente receber novos valores para produzir a sequência de sons. O número 16514 é o endereço do carácter seguinte ao comando REM na primeira linha do programa (no caso, o carácter interpretado é o primeiro 'Y'). Nos endereços 16515 e 16519, que atualmente correspondem ao segundo e quarto Y, o programa BASIC inserirá os valores D e C para comunicá-los ao PSG. Se o programa é listado depois de rodado, o comando REM será alterado justamente no segundo e quarto 'Y' com outros caracteres ou comandos. Assim, por curiosidade, compare os novos caracteres gerados no comando REM com a tabela localizada no APÊNDICE A do seu manual.

OPERAÇÕES INTERNAS DO PSG

É necessário entender um pouco das operações internas do PSG para conseguir obter todas as vantagens e facilidades que o PSG pode proporcionar. A exata função de cada um dos registros internos e os números que devem ser carregados para ativar o efeito desejado são o que agora se descreverá.

REGISTROS D0 E D1

Estes registros regulam a frequência do canal A do PSG. O registro 0 faz o ajuste grosso enquanto o 1 faz o fino ajuste. A combinação dos dois registros irá, de acordo com a fórmula abaixo, proporcionar o valor da frequência de uma nota musical:

$$\text{freq} = 1625000 / (16 * (256 * D1 + D0))$$

sendo D0 e D1 os valores dos conteúdos dos registros D0 e D1 respectivamente. O registro D0 deve somente conter valores inteiros (sem parte fracionária) da faixa de 0 a 255, enquanto o registro D1 deverá conter também somente valores inteiros mas somente na faixa de 0 até 15. Se ambos valores forem zero não será gerado som no canal A. Para isso seria melhor utilizar o registro 0. A nota mais aguda será gerada fazendo-se D0=1 e D1=0 gerando então 10kHz (completamente inaudível) ou, se fizermos D1=1 e D0=132, teremos uma nota de 261 Hz.

REGISTROS D2 E D3

Estes operam no canal B exatamente da mesma maneira com que os registros D0 e D1 operam no canal A.

REGISTROS D4 E D5

Este operam exatamente como no caso anterior, mas no canal C.

REGISTRO D6

Este registro controla o período do ruído gerador nos três canais, quando habilitado, isto é, quando o registro D7 está atuante. Este registro pode ter seus valores entre 1 e 31. O maior valor corresponderá a um ruído forte (como uma explosão) e o menor corresponderá a um ruído baixo (como um vazamento de gás).

REGISTRO D7

Este registro é usado para habilitar e desabilitar ruídos e tons.

No canal A o ruído é desabilitado fazendo-se D7=8
No canal B o ruído é desabilitado fazendo-se D7=16
No canal C o ruído é desabilitado fazendo-se D7=32
No canal A o tom é desabilitado fazendo-se D7=1
No canal B o tom é desabilitado fazendo-se D7=2
No canal C o tom é desabilitado fazendo-se D7=4

Deste modo, fazendo-se D7=0 habilitamos ruído e tons para os três canais. Similarmente, carregando-se 7 (isto é, 1 + 2 + 4) habilitamos somente o ruído para os três canais, enquanto que carregando-se 56 (isto é, 8 + 16 + 32) habilitamos somente tons nos três canais.

REGISTROS D8, D9 E D10

Estes três registros regulam separadamente os volumes dos canais A, B e C respectivamente. O mínimo volume é dado pelo valor zero(0) e o máximo pelo 15. Carregando-se 16, você estará utilizando a modulação de volume controlado pelos próximos registros.

REGISTROS D11 E D12

Estes registros controlam a faixa de variação de volume nos canais habilitados pelo modo variável. O período de variação é dado pela equação:

$$p = 162500 / (256 * (256 * D12 + D11))$$

Os valores de D11 e D12 de D11 e D12 podem ambos ter valores entre 0 e 255. Se ambos forem carregados com 255, uma variação muito lenta de volume poderá ser observada com período de variação de 10 segundos aproximadamente.

REGISTRO D13

Este registro controla o tipo de modulação do volume a ser produzido. A operação de D13 é melhor descrita através da listagem de todos os efeitos possíveis das diferentes maneiras de se carregar D13:

D13= 0	Descida simples
D13= 6	Subida simples, seguida de silêncio
D13= 8	Dente de serra em descida contínua
D13=10	Subida e descida contínua
D13=11	Descida simples seguida pelo volume máximo
D13=12	Dente de serra em subida contínua
D13=13	Subida simples seguida do volume máximo

D13=14 Como i0, só que inicia com uma subida do volume.

Com o uso do PSG você notará que, provavelmente, o mais útil será D13=0, pois você, neste valor, poderá utilizar o computador como piano ou para explosões etc...

ALGUNS PROGRAMAS UTEIS

A seguir serão apresentadas uma sequência de programas úteis que gerarão sons que você poderá experimentar ou usá-los em seus próprios programas. Deve-se, no entanto, sempre considerar a existência das linhas 1 a 10, pois como já foi mencionado, elas são vitais para o funcionamento do PSG.

LIMPADOR DOS REGISTROS

Este segmento de programa carrega todos os registros com zero, iniciando a operação do PSG.

```
100 LET C=0
110 FOR D = 0 TO 13
120 GOSUB 3
130 NEXT D
```

Note que o comando LET C=0 não foi introduzido dentro do comando FOR. Isto ocorre em virtude de a sub-rotina da linha 3 não alterar este valor. Ela apenas o lê.

QUEDA DE BOMBA

Este programa provoca um decaimento de frequência produzindo um efeito sonoro semelhante à queda de uma bomba. O programa abaixo virá acompanhado de frases que apenas o ajudarão a entender o que o programa está fazendo (não tente copiá-las). Para melhor execução deste programa procure utilizar o modo FAST.

```
200 LET D=7
210 LET C=62 (somente habilita A, 62=2+4+8+16+32)
220 GOSUB 3
230 LET D=8
240 LET C=15 (canal A em volume máximo)
250 GOSUB 3
260 LET D=0 (frequência do canal A)
270 FOR C=48 TO 192
280 GOSUB 3
290 NEXT C (para diminuir a frequência)
300 LET D=7
310 LET C=63
320 GOSUB 3 (para silenciar o canal A)
```

EXPLOSAO

Este efeito sonoro fica muito bem quando seguido imediatamente apds a queda da bomba! O modo FAST também é ideal para este caso.

```
400 LET D=6
410 LET C=31
420 GOSUB 3
430 LET D=7
440 LET C=7 (habilita somente o ruído)
450 GOSUB 3
460 LET C = 16 (para a modulação de volume)
470 FOR D=9 TO 10 (isto é, os três canais)
480 GOSUB 3
490 NEXT D
500 LET D=12 (período da modulação)
510 LET C=56 (um decaimento lento)
520 GOSUB 3
530 LET D=13
540 LET C=0 (uma simples explosão)
550 GOSUB 3
```

TIRO DE ESPINGARDA

Para realizar este efeito, basta trocar no programa acima o valor de C na linha 410 para 1 e o valor de C na linha 510 para 16.

TIRO LASER

Este programa varre uma frequência de ruído, sendo que o volume decai, dando a impressão de um tiro laser. Este som será também melhor produzido no modo FAST.

```
600 LET C=0
610 FOR D=9 TO 10
620 GOSUB 3
630 NEXT D
640 LET D=7
650 LET C=55 (ruído somente no canal A)
660 GOSUB 3
670 LET D=8
680 LET C=16 (modulação de volume)
690 GOSUB 3
700 LET D=12
710 LET C=15
720 GOSUB 3
730 LET D=13
740 LET C=0 (decaimento simples)
750 GOSUB 3
760 LET D=6
770 FOR C=1 TO 31 STEP 3
780 GOSUB 3
```

```
790 NEXT C
800 STOP
```

ASSOBIO

Este programa deverá ser rodado em FAST

```
600 LET D=6
610 LET C=1 (para gerar um assobio)
620 GOSUB 3
630 LET D=7
640 LET C=46 (canal A com tons e B com ruído)
650 GOSUB 3
660 LET D=8
670 LET C=15 (volume completo em A)
680 GOSUB 3
690 LET D=9
700 LET C=46
710 GOSUB 3
720 LET D=0
730 FOR C=100 TO 30 STEP -3
740 GOSUB 3
750 NEXT C
760 LET D=7
770 LET C=63 (silêncio)
780 GOSUB 3
790 FOR T=1 TO 5 (para uma pausa)
800 NEXT T
810 LET C=46
820 GOSUB 3
830 LET D=0
840 FOR C=30 TO 60 STEP -3
850 GOSUB 3
860 NEXT C
870 FOR C=30 TO 100 STEP 3
880 GOSUB 3 (D e ainda 0)
890 NEXT C
900 LET C=0
910 FOR D=8 TO 10
920 GOSUB 3
930 NEXT D
940 STOP
```

ORGÃO ELETRÔNICO

Este programa utiliza as teclas 1 até 0 de seu TK-85 como teclado do órgão. A nota se manterá constante enquanto você estiver com o dedo pressionando a tecla. Se você retirar o dedo da tecla, a nota cessará exatamente como num órgão real. Rode este programa no modo FAST.

```
100 LET C=0
110 FOR D=1 TO 11
120 GOSUB 3 (limpa todos os registros)
130 NEXT D
140 LET D=7
150 LET C=62 (tom somente no canal A)
160 GOSUB 3
170 LET D=8
180 LET C=16
190 GOSUB 3
200 LET D=12
210 LET C=30
220 GOSUB 3
230 DIM B(10)
240 LET B(1)=180
250 LET B(2)=252
260 LET B(3)=225
270 LET B(4)=200
280 LET B(5)=189
290 LET B(6)=168
300 LET B(7)=150
310 LET B(8)=134
320 LET B(9)=126
330 LET B(10)=112
340 LET G=(CODE INKEY$)-27
350 IF G(1 OR G)>10 THEN GOTO 340
360 LET D=0
370 LET C=B(G)
380 GOSUB 3
390 LET D=13
400 LET C=0
410 GOSUB 3
420 GOTO 340
```

O PROGRAMA EM LINGUAGEM DE MÁQUINA

O programa em Linguagem de Máquina localizado na linha REM utiliza os dados passados através dos comandos F0KÉ's localizados nas linhas 3 e 4 da sub-rotina. Existem 2 portas I/O de acesso ao PSG pelo Z80 do TK-85: a porta endereçada em CF (hexadecimal, normalmente escrita por OCFH), na qual são indicados os registros a serem alterados, e a porta endereçada em UFH, com a informação para o registro selecionado.

A versão assembler do programa em Linguagem de Máquina carregando, por exemplo, no registro 7 com 63 (3FH) é o seguinte

```
16514 3E LD A,07
16515 07
16516 D3 OUT (OCFH),A
16517 CF
16518 3E LD A,3FH
16519 3F
16520 D3 OUT (OFH),A
16521 0F
16522 C9 RET
```

Você poderá conferir o código hexadecimal com as palavras existentes na linha 1 do programa através da tabela existente no Apêndice A do Manual de BASIC do TK-85.

...the first of these is the fact that the ...

...the second of these is the fact that the ...

...the third of these is the fact that the ...

...the fourth of these is the fact that the ...

...the fifth of these is the fact that the ...

...the sixth of these is the fact that the ...

...the seventh of these is the fact that the ...

...the eighth of these is the fact that the ...

...the ninth of these is the fact that the ...

...the tenth of these is the fact that the ...

...the eleventh of these is the fact that the ...

...the twelfth of these is the fact that the ...

...the thirteenth of these is the fact that the ...

...the fourteenth of these is the fact that the ...

...the fifteenth of these is the fact that the ...

...the sixteenth of these is the fact that the ...

...the seventeenth of these is the fact that the ...

...the eighteenth of these is the fact that the ...